Math 276
Calculus III

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 12
Partial Derivatives

Initializations

```
> restart;
  with(plots):
>
```

12.1 Partial Derivatives of Maple Expressions

In Maple, derivatives of expressions containing more than one variable, are automatically considered to be partial derivatives and can be computed just like we have computed derivatives in the past.

Examples

Example 12.1.1

Compute the first and second partial derivatives of $x^3 + \sin(x\,y) + x^2\,y$.

```
> expression:=x^3+sin(x*y)+x^2*y;
```

$$expression := x^3 + \sin(x\,y) + x^2\,y \quad \textbf{(2.1.1.1)}$$

```
> Diff(expression, x)=diff(expression, x);
```

$$\frac{\partial}{\partial x}\left(x^3 + \sin(x\,y) + x^2\,y\right) = 3\,x^2 + \cos(x\,y)\,y + 2\,x\,y \quad \textbf{(2.1.1.2)}$$

```
> Diff(expression, y)=diff(expression, y);
```

$$\frac{\partial}{\partial y}\left(x^3 + \sin(x\,y) + x^2\,y\right) = \cos(x\,y)\,x + x^2 \quad \textbf{(2.1.1.3)}$$

```
> Diff(expression, x$2)=diff(expression, x$2);
```

$$\frac{\partial^2}{\partial x^2}\left(x^3 + \sin(x\,y) + x^2\,y\right) = 6\,x - \sin(x\,y)\,y^2 + 2\,y \quad \textbf{(2.1.1.4)}$$

```
> Diff(expression, x, y)=diff(expression, x, y);
```

$$\frac{\partial^2}{\partial y\,\partial x}\left(x^3 + \sin(x\,y) + x^2\,y\right) = -\sin(x\,y)\,x\,y + \cos(x\,y) + 2\,x \quad \textbf{(2.1.1.5)}$$

Observe that in the formula above we first differentiate with respect x then with respect to y. In the formula below we first differentiate with respect y then with respect to x.

```
> Diff(expression, y, x)=diff(expression, y, x);
```

$$\frac{\partial^2}{\partial x\,\partial y}\left(x^3 + \sin(x\,y) + x^2\,y\right) = -\sin(x\,y)\,x\,y + \cos(x\,y) + 2\,x \quad \textbf{(2.1.1.6)}$$

```
> Diff(expression, y$2)=diff(expression, y$2);
```

$$\frac{\partial^2}{\partial y^2}\left(x^3+\sin(x\,y)+x^2\,y\right)=-\sin(x\,y)\,x^2 \qquad (2.1.1.7)$$

```
>
```

12.2 Partial Derivatives of Maple Functions

Just like derivatives of functions of one variable, it is possible to render the derivatives of a Maple function of several variables as a Maple function in stead of a Maple expression. This is particularly handy when the derivative needs to be evaluated for multiple values of the arguments.

Examples

Example 12.2.1

Compute the first and second derivatives of $f(x, y) = x^3 + \sin(x\,y) + x^2\,y$ as Maple functions. Compute $f_{xy}(2, 3)$.

```
> f:=(x, y)->x^3+sin(x*y)+x^2*y;
```

$$f := (x, y) \rightarrow x^3+\sin(x\,y)+x^2\,y \qquad (3.1.1.1)$$

```
> Diff(f, x)=D[1](f);
```

$$\frac{\partial}{\partial x}\,f=\left((x, y)\rightarrow 3\,x^2+\cos(x\,y)\,y+2\,x\,y\right) \qquad (3.1.1.2)$$

```
> Diff(f, y)=D[2](f);
```

$$\frac{\partial}{\partial y}\,f=\left((x, y)\rightarrow \cos(x\,y)\,x+x^2\right) \qquad (3.1.1.3)$$

```
> Diff(f, x$2)=D[1$2](f);
```

$$\frac{\partial^2}{\partial x^2}\,f=\left((x, y)\rightarrow 6\,x-\sin(x\,y)\,y^2+2\,y\right) \qquad (3.1.1.4)$$

```
> Diff(f, x, y)=D[1, 2](f);
```

$$\frac{\partial^2}{\partial y\,\partial x}\,f=\left((x, y)\rightarrow -\sin(x\,y)\,y\,x+\cos(x\,y)+2\,x\right) \qquad (3.1.1.5)$$

```
> Diff(f, y, x)=D[2, 1](f);
```

$$\frac{\partial^2}{\partial x\,\partial y}\,f=\left((x, y)\rightarrow -\sin(x\,y)\,y\,x+\cos(x\,y)+2\,x\right) \qquad (3.1.1.6)$$

```
> Diff(f, y$2)=D[2$2](f);
```

$$\frac{\partial^2}{\partial y^2}\,f=\left((x, y)\rightarrow -\sin(x\,y)\,x^2\right) \qquad (3.1.1.7)$$

Finally, we evaluate $f_{xy}(2, 3)$.

```
> D[1,2](f)(2,3);
```

$$-6\sin(6)+\cos(6)+4 \qquad (3.1.1.8)$$

```
>
```

12.3 Partial Derivatives as Slopes of Tangent Lines

Examples

Example 12.3.1

Find and display the tangent line at the point $(1, 1, 2)$ to the curve of intersection of the paraboloid $z = f(x, y) = 4 - x^2 - y^2$ and the plane $y = 1$.

Solution

The intersection $z = 4 - x^2 - y^2$ and $y = 1$ is the space curve

$$\mathbf{r}_1(x) = \langle x, 1, 3 - x^2 \rangle$$

The slope of this tangent line is given by $m = f_x(1, 1)$. Consequently the equations of this line are

$$z - 2 = f_x(1, 1)(x - 1)$$
$$y = 1$$

Hence, a vector equation of this tangent line is given by

$$\mathbf{r}_2(x) = \langle x, 1, f_x(1, 1)(x - 1) + 2 \rangle$$

```
> f:=(x,y)->4-x^2-y^2;
  r1:=<x,1,3-x^2>:
  r2:=<x,1,D[1](f)(1,1)*(x-1)+2>:
  r1, r2;
```

$$f := (x, y) \rightarrow 4 - x^2 - y^2$$

$$\begin{bmatrix} x \\ 1 \\ 3 - x^2 \end{bmatrix}, \begin{bmatrix} x \\ 1 \\ -2x + 4 \end{bmatrix} \qquad \textbf{(4.1.1.1)}$$

Next we create the images of the paraboloid $z = 4 - x^2 - y^2$, the plane $y = 1$, the curve of intersection, and the tangent line.

```
> p1:=plot3d(f(x,y), x=-3..3, y=-3..3):
  p2:=plot3d([x, 1, z], x=-2..2, z=0..4, color=red):
  p3:=spacecurve(r1, x=-2..2, color=black, thickness=4):
  p4:=spacecurve(r2, x=-2..2, color=blue, thickness=4):
> display([p1, p2, p3, p4], axes=boxed, view=[-2..2, -2..2,
  0..4]);
```



x
2
1
0
-1
-2
2
1
0
1
2
4

>